# Vektoreita GeoGebralla

Vektoreilla voi laskea joko komentopohjaisesti esim. CAS-ikkunassa tai piirtämällä piirtoikkunassa. Ensimmäisen tavan etuna on, että laskujen tueksi muodostuu kuva. Tästä on varmasti apua sekä käsitteiden oppimisessa että ratkaisun tarkistamisessa.

## Yleistä vektoreista GeoGebralla

GeoGebrassa pisteet nimetään isoilla ja vektorit pienillä kirjaimilla. Esim. CAS ikkunassa a:=(1,2) määrittelee vektorin. Vapaata vektoria voi liikuttaa ottamalla siitä kiinni vektorin alku- tai keskiosasta. Kärjen tuntumasta kiinni ottaminen lähtee muuttamaan vektoria. Geogebralla laskettaessa pisteillä voi laskea kuten paikkavektoreita. Vektoria ei pysty liikuttamaan, jos vektori on tuotettu laskutoimituksella. Tällöin sen arvo on kiinnitetty. Vektorista voi kuitenkin tuottaa haluamansa edustajan komennolla Siirto[*vektori*, *alkupiste*] tai työkaluvalikosta painikkeella 

.

Kahden vektorin pistetuloa merkitään kertomerkillä. Pistetuloa varten on myös oma komentonsa, Pistetulo[*vektori*, *vektori*]. Vektoreita voi tuottaa suoraan piirtoikkunassa. Työkalupalkissa on kaksi vektoreihin liittyvää toimintoa, *Vektori pisteestä pisteeseen* ja *Vektori: alkupiste ja vektori*. Nämä voi tehdä myös komentoina joko CAS-ikkunassa tai syöttökentässä. Ensimmäinen tulee komennolla Vektori[*Piste, Piste*] ja jälkimmäinen Siirto[*vektori, alkupiste*]. Jos tehtävän tekee suoraan piirtämällä, piirtämisen vaiheet saa näkyviin valitsemalla näytä valikosta konstruktion vaiheet. Tästä saattaisi olla apua kokeissa, kun on tarvetta kertoa, mitä tuli tehtyä.




Muita vektoreihin liittyviä komentoja saa Komennot listasta näkyviin. Näitä on esimerkiksi

| Komento | Huom. |
|---|---|
| Ristitulo[vektori, vektori] | Saa myös CAS-ikkunan virtuaalinäppäimistöstä a⊗b |
| Yksikkövektori[ <Objekti> ] | Esim. suoran suuntavektori |
| Yksikkövektori[ <Vektori> ] | CAS-puolella |
| Normaalivektori[ <Objekti> ] | Objektina voi olla suora, jana, vektori tai taso |
| KohtisuoraYksikkövektori[ < Objekti>] | Objektina voi olla suora, jana, vektori tai taso |

## Esimerkkitehtäviä

CAS-ikkunassa asioita määritellään := merkillä, kun syöttökentässä riittää = merkki. Vektori $\bar{a} = 4\vec{i} - 2\vec{j} + 6\vec{k}$ syötetään GeoGebrassa muodossa $a := (4, -2, 6)$. Näissä esimerkeissä hyödynnän sitä, että pisteillä voi laskea kuten vektoreilla. Pedanttisempi henkilö voisi paikkavektoreita laskea ensin komennolla Vektori[(0,0,0), A], kun esimerkeissä saan saman komennolla a:=A. Geogebran komennoissa pitää olla tarkkana, esim. Siirto-komennossa tulee ensimmäisen parametrin olla vektori ja toisen piste. Esimerkeissä painotetaan ratkaisun tekemistä CAS-ikkunassa. Geogebran etuna on, että lasketun asian pystyy varmistamaan piirtoikkunassa, 2D tai 3D.

1) Pisteet $A=(-3,4), B=(2,3)$.

a) Laske $\bar{a}=\overline{AB}$

b) Laske $|\bar{a}|$.

c) Laske janan $AB$ keskipisteen C paikkavektori

d) Piste P jakaa janan AB suhteessa 2:3. Määritä P:n paikkavektori.

A := (-3, 4)

$\rightarrow \quad \mathbf{A} := (-3, 4)$

B := (2, 3)

$\rightarrow \quad \mathbf{B} := (2, 3)$

a:=B-A

$\rightarrow \quad \mathbf{a} := \begin{pmatrix} 5 \\ -1 \end{pmatrix}$

Vektori[ alkupiste, loppupiste]

u:=Vektori[A,B]

$\rightarrow \quad \mathbf{u} := \begin{pmatrix} 5 \\ -1 \end{pmatrix}$

Laskennan tuloksena syntynyttä vektoria ei voi liikuttaa. Vektorin voi kopioida alkamaan pisteestä A työkalupalkin komennolla *vektori, alkupiste ja vektori*.




Pituus[a]

$\rightarrow \quad \sqrt{26}$

oc:=(A+B)/2

$\rightarrow \quad \mathbf{oc} := \begin{pmatrix} -\frac{1}{2} \\ \frac{7}{2} \end{pmatrix}$

cc:=A+a/2

$\rightarrow \quad \mathbf{cc} := \begin{pmatrix} -\frac{1}{2} \\ \frac{7}{2} \end{pmatrix}$

op:=(3A+2B)/(2+3)

$\rightarrow \quad \mathbf{op} := \begin{pmatrix} -1 \\ \frac{18}{5} \end{pmatrix}$

pp:=A+2/5*a

$\rightarrow \quad \mathbf{pp} := \begin{pmatrix} -1 \\ \frac{18}{5} \end{pmatrix}$

Paikkavektorista pisteeksi. Kuvassa Geogebra esittää pisteenä.

P:=pp

$\rightarrow \quad \mathbf{P} := \begin{pmatrix} -1 \\ \frac{18}{5} \end{pmatrix}$

C:=oc

$\rightarrow \quad \mathbf{C} := \begin{pmatrix} -\frac{1}{2} \\ \frac{7}{2} \end{pmatrix}$



2) $\bar{a} = 3\vec{i} - 2\vec{j} - 4\vec{k}$ ja $\bar{b} = \vec{i} + 2\vec{j} + 2\vec{k}$. Laskut etenevät ensin vasemman puolen sarakkeessa.

a:=(3,-2,-4)

$\rightarrow \; \mathbf{a} := \begin{pmatrix} 3 \\ -2 \\ -4 \end{pmatrix}$

b:=(1,2,2)

$\rightarrow \; \mathbf{b} := \begin{pmatrix} 1 \\ 2 \\ 2 \end{pmatrix}$

pituudet

Pituus[a]

$\rightarrow \; \sqrt{29}$

Pituus[b]

$\rightarrow \; 3$

Pistetulo

a*b

$\rightarrow \; -9$

Pistetulo[a,b]

$\rightarrow \; -9$

acosd(Pistetulo(a,b)/(Pituus(a)*Pituus(b)))

$\checkmark \; \text{acosd}\left(\frac{\text{Pistetulo}[a, b]}{\text{Pituus}[a] \; \text{Pituus}[b]}\right)$

acosd(Pistetulo(a,b)/(Pituus(a)*Pituus(b)))

$\approx \; 123.85°$

Vektoriprojektion voisi määrittää piirtämälläkin

(a*b)/(b*b)*b

$\rightarrow \; \begin{pmatrix} -1 \\ -2 \\ -2 \end{pmatrix}$

Huom! Painikkeet, tarkka arvo, likiarvo ja tarkista lauseke, CAS työkaluissa. Asteen palauttavat käänteiset trigonometriset funktiot loppuvat d-kirjaimeen.

3) Piste $A = (2,1,-3)$ on vektorin $\bar{c}$ alkupiste. Määritä loppupiste B, kun $\bar{c} = 3\vec{i} - 2\vec{j} - 4\vec{k}$.

Suoralla laskulla:

B:=(2,1,-3)+(3,-2,-4)

$\rightarrow \; \mathbf{B} := (5, -1, -7)$

Vaihtoehtoinen tapa vieressä.

1 B:=(x,y,z)

$\rightarrow \; \mathbf{B} := (x, y, z)$

2 A:=(2,1,-3)

$\rightarrow \; \mathbf{A} := (2, 1, -3)$

3 c:=(3,-2,-4)

$\rightarrow \; \mathbf{c} := \begin{pmatrix} 3 \\ -2 \\ -4 \end{pmatrix}$

4 Ratkaise[c=B-A,{x,y,z}]

$\rightarrow \; \{\{x = 5, y = -1, z = -7\}\}$



4) Pisteestä $A = (1, -1, 0)$ siirrytään 9 pituusyksikköä $\vec{\imath} - 2\vec{\jmath} + 2\vec{k}$ suuntaan pisteeseen B ja siitä edelleen 10 pituusyksikköä vektorin $\overrightarrow{3\imath} - 4\vec{k}$ suuntaan pisteeseen C. Määritä pisteen C koordinaatit. ( s2013 5)

| | |
|---|---|
| 1 | **Yksikkövektori[ <Vektori> ] komennolla** |
| 2 | C:=(1,-1,0)+9*Yksikkövektori[(1,-2,2)]+10*Yksikkövektori[(3,0,-4)] <br> → $C := (10, -7, -2)$ |

5) Vektori $\vec{a} = \vec{\imath} - 2\vec{\jmath} + 4\vec{k}$ ja $\vec{b} = 5\vec{\imath} - 10\vec{\jmath} + (t + 1)\vec{k}$. Millä $t$:n arvoilla vektorit ovat yhdensuuntaisia?

| | |
|---|---|
| 1 | a:=(1,-2,4) <br> → $a := \begin{pmatrix} 1 \\ -2 \\ 4 \end{pmatrix}$ |
| 2 | b:=(5,-10,t+1) <br> → $b := \begin{pmatrix} 5 \\ -10 \\ t+1 \end{pmatrix}$ |
| 3 | Ratkaise[b=r*a,{r,t}] <br> → $\{\{r = 5, t = 19\}\}$ |
| 4 | Vastaus: Kun t = 19 |

6) Jaa vektori $\vec{\imath} + 7\vec{\jmath}$ vektoreiden $\vec{a} = 2\vec{\imath} + 3\vec{\jmath}$ ja $\vec{b} = -7\vec{\imath} + 6\vec{\jmath}$ suuntaisiin komponentteihin.
Teen tämän ensin piirtämällä. Määritin vektorit ja niiden loppupisteet, kun alkupiste origossa. Piirsin vektorin a suuntaisen suoran pisteen C kautta. Määritin leikkauspisteen D, jolloin halutut komponentit ovat vektorit u ja v. CAS puolella saan niiden tarkat arvot, jos kertoimet ovat murtolukuja.




| | |
|---|---|
| u | → $\begin{pmatrix} -\frac{7}{3} \\ 2 \end{pmatrix}$ |
| v | → $\begin{pmatrix} \frac{10}{3} \\ 5 \end{pmatrix}$ |

Toinen tapa, CAS-puolella vektoriyhtälöllä

Ratkaise[c=r*a+s*b,{r,s}]

$\rightarrow \left\{\left\{r = \frac{5}{3}, s = \frac{1}{3}\right\}\right\}$

5/3*a

$\rightarrow \begin{pmatrix} \frac{10}{3} \\ 5 \end{pmatrix}$

1/3*b

$\rightarrow \begin{pmatrix} -\frac{7}{3} \\ 2 \end{pmatrix}$

7) Vektori $\vec{a} = 4\vec{i} - 5\vec{j} + 3\vec{k}$ ja $\vec{b} = 2\vec{i} + \vec{j} - 2\vec{k}$. Esitä vektori $\vec{a}$ summana, joista toinen komponentti on $\vec{b}$:n suuntainen, toinen kohtisuoraan $\vec{b}$:tä vasten.

Piirtämällä:
Määritetään vektorit $\vec{a}$ ja $\vec{b}$ sekä loppupisteet, kun alkupiste origossa. Piirretään suora, joka on $\vec{b}$:n suuntainen ja kulkee origon kautta. Piirretään tätä suoraa vasten kohtisuora taso, joka kulkee vektorin $\vec{a}$ loppupisteen kautta. Määritetään suoran ja tason leikkauspiste C. Halutut komponentit ovat $\overrightarrow{OC}$ ja $\overrightarrow{CA}$. Kokeile itse vektoriprojektion avulla.




u

6 $\rightarrow \begin{pmatrix} -\frac{2}{3} \\ -\frac{1}{3} \\ \frac{2}{3} \end{pmatrix}$

v

7 $\rightarrow \begin{pmatrix} \frac{14}{3} \\ -\frac{14}{3} \\ \frac{7}{3} \end{pmatrix}$

8) Ovatko pisteet P=(4,1,-2) ja Q=(0,2,4) pisteiden A=(1,1,1) ja B(-1,1,3) määräämällä suoralla?

a:=(1,1,1)+t*((-1,1,3)-(1,1,1))

$$\rightarrow \quad \mathbf{a} := \begin{pmatrix} 1 - 2\,t \\ 1 \\ 1 + 2\,t \end{pmatrix}$$

a=(4,1,-2)

Ratkaise: $\left\{ t = -\frac{3}{2} \right\}$

**Eli piste P on suoralla**

a=(3,1,3)

Ratkaise: $\{\}$

Eli piste Q ei ole suoralla

9) Pisteiden $A(2,01)$ ja $B(3,1,3)$ yhdysjanan keskipisteen kautta asetetaan taso, joka on kohtisuorassa yhdysjanaa vasten. Missä pisteessä taso leikkaa y-akselin?
Piirretään pisteet. Jana pisteiden kautta. Keskipistetyökalulla keskipiste. Taso keskipisteen kautta, kohtisuora janaa vasten. Piirrä suora y-akselille. Määritä tason ja y-akselin leikkauspiste F.




10) Onko piste (-1,-3,6) pisteiden (1,3,2), (-2,1,5) ja (2,-1,3) määräämässä tasossa.

11)

A:=(1,3,2)

→ $A := (1, 3, 2)$

B:=(-2,1,5)

→ $B := (-2, 1, 5)$

C:=(2,-1,3)

→ $C := (2, -1, 3)$

Taso[A,B,C]

→ $5x + 3y + 7z = 28$

5x + 3y + 7z = 28

Sijoita, x=-1,y=-3,z=6: $28 = 28$

Joten piste on tason piste. Tason yhtälön saa myös piirtämällä tason.

12)

Suoran pyramidin pohjana on suunnikas $ABCD$

a) Määritä pisteen $C$ paikka, kun $A = (-2, 1, 0)$, $B = (1, 3, 0)$ ja $D = (-4, 2, 2)$.

b) Määritä pyramidin tilavuus, kun sen huipun z-koordinaatti on 5.

Tämän voisi tehdä puhtaasti piirtämällä. Mutta nyt vektoreilla. Suunnikkaan sivuina ovat vektori AB ja AD. Lasketaan pisteen C paikkavektori c.

1 → $A := (-2, 1, 0)$

2 B:=(1,3,0)

→ $B := (1, 3, 0)$

3 D:=(-4,2,2)

→ $D := (-4, 2, 2)$

4 s_1:=D-A

→ $s_1 := \begin{pmatrix} -2 \\ 1 \\ 2 \end{pmatrix}$

5 s_2:=B-A

→ $s_2 := \begin{pmatrix} 3 \\ 2 \\ 0 \end{pmatrix}$

6 c:=A+s_1+s_2

→ $c := \begin{pmatrix} -1 \\ 4 \\ 2 \end{pmatrix}$

7 Vastaus: C = (-1,2,2)



| | |
|---|---|
| 9 | Suunnikkaan keskipiste E |
| 10 | E:=A+(s_1+s_2)/2 <br> $\rightarrow$ $E := \left(-\frac{3}{2}, \frac{5}{2}, 1\right)$ |
| 11 | Tason ABD vastaan normaalivektori, ja suora pisteen E kautta, suunta n |
| 12 | n:=Ristitulo[s_1,s_2] <br> $\rightarrow$ $n := \begin{pmatrix} -4 \\ 6 \\ -7 \end{pmatrix}$ |
| 13 | E+t*n <br> $\rightarrow$ $\left(-\frac{3}{2} - 4\,t, \frac{5}{2} + 6\,t, 1 - 7\,t\right)$ |
| 14 | Leikkaa tason z = 5, sijoitetaan suoran yhtälöstä x, y, ja z |
| 15 | 1-7t=5 <br> Ratkaise: $\left\{t = -\frac{4}{7}\right\}$ |
| 16 | Jolloin pyramidin huippu on |
| 17 | F:=Sijoita[((-3) / 2 - 4t, 5 / 2 + 6t, 1 - 7t),{t = (-4) / 7}] <br> $\rightarrow$ $F := \left(\frac{11}{14}, -\frac{13}{14}, 5\right)$ |
| 18 | Pyramidin korkeus on pisteiden F ja E etäisyys |

| | |
|---|---|
| 19 | h:=Etäisyys[E,F] <br> $\rightarrow$ $h := 4 \cdot \frac{\sqrt{101}}{7}$ |
| 20 | Pohjan ala on vektorin n pituus |
| 21 | ALA:=Pituus[n] <br> $\rightarrow$ $ALA := \sqrt{101}$ |
| 22 | V:=ALA*h/3 <br> $\rightarrow$ $V := \frac{404}{21}$ |

Varmistus piirtämällä, pisteet ovat jo 3D näkymässä




Piirretyn pyramidin tilavuus CAS ikkunassa

V:=ALA*h/3

$\rightarrow$ $\mathbf{V} := \frac{\mathbf{404}}{\mathbf{21}}$

e

$\rightarrow$ $\frac{\mathbf{404}}{\mathbf{21}}$

13) Suora $l_1$ kulkee pisteiden $A = (1,1,4)$ ja $B = (1,2,3)$ kautta. Suora $l_2$ kulkee pisteiden $C = (-2,3,2)$ ja $B = (0,2,5)$ kautta. Määritä suorien välinen etäisyys. Idea: jaetaan vektori CB kolmeen komponenttiin.

| | |
|---|---|
| 1 | A:=(1,1,4)<br>→ $\mathbf{A := (1,1,4)}$ |
| 2 | B:=(1,2,3)<br>→ $\mathbf{B := (1,2,3)}$ |
| 3 | s_1:=B-A<br>→ $\mathbf{s_1 := \begin{pmatrix} 0 \\ 1 \\ -1 \end{pmatrix}}$ |
| 4 | f:=Suora[A,B]<br>→ $\mathbf{f : X = (1,1,4) + \lambda\ (0,1,-1)}$ |
| 5 | C:=(-2,3,2) |
| 6 | D:=(0,2,5)<br>→ $\mathbf{D := (0,2,5)}$ |
| 7 | s_2:=D-C<br>→ $\mathbf{s_2 := \begin{pmatrix} 2 \\ -1 \\ 3 \end{pmatrix}}$ |
| 8 | n:=Ristitulo[s_1,s_2]<br>→ $\mathbf{n := \begin{pmatrix} 2 \\ -2 \\ -2 \end{pmatrix}}$ |
| 9 | Suorat eivät ole yhdensuuntaisia |
| 10 | Ratkaise[B-C=r*s_1+s*s_2+t*n,{r,s,t}]<br>→ $\left\{\left\{\mathbf{r = 1, s = 1, t = \frac{1}{2}}\right\}\right\}$ |
| 11 | suorien välinen etäisyys on komponentin t*n Pituus |
| 12 | Pituus[1/2*n]<br>→ $\sqrt{3}$ |
| 13 | sqrt(3)<br>≈ $\mathbf{1.73}$ |




Kuvaa varten on ensin laskettu vektori $\bar{n}/2$ ja sitten työkalulla vektori alkupiste ja vektori piirretty komponenttivektorit alkaen pisteestä C. Suuntavektoreiden kertoimet olivat ykkösiä, joten niiden suuntaisia komponentteja ei tarvinnut erikseen laskea.

14) Tasosta tiedetään pisteen A paikkavektori sekä tason virittävät vektorit $\bar{s_1}$ ja $\bar{s_2}$. Määritä tason vektorimuotoinen ja normaalimuotoinen yhtälö.

oa:=(1,3,-2)

$$\rightarrow \quad \mathbf{oa} := \begin{pmatrix} 1 \\ 3 \\ -2 \end{pmatrix}$$

s_1:=(2,-1,4)

$$\rightarrow \quad \mathbf{s_1} := \begin{pmatrix} 2 \\ -1 \\ 4 \end{pmatrix}$$

s_2:=(3,1,5)

$$\rightarrow \quad \mathbf{s_2} := \begin{pmatrix} 3 \\ 1 \\ 5 \end{pmatrix}$$

Tason piste A paikkavektori oa

Tason suuntavektorit s_1 ja s_2, tason T yhtälö on:

T:(x,y,z)=oa+t*s_1+r*s_2

$\rightarrow$ $\mathbf{T : (x, y, z) = (3\,r + 2\,t + 1, r - t + 3, 5\,r + 4\,t - 2)}$

Ratkaistaan parametrit r ja t sekä z, jolloin saamme tason normaalimuotoisen yhtälön

Ratkaise[T,{z,r,t}]

$\rightarrow$ $\left\{\left\{z = \frac{9}{5}\,x - \frac{2}{5}\,y - \frac{13}{5}, r = \frac{1}{5}\,x + \frac{2}{5}\,y - \frac{7}{5}, t = \frac{1}{5}\,x - \frac{3}{5}\,y + \frac{8}{5}\right\}\right\}$

(z = 9 / 5 x - 2 / 5 y - 13 / 5)*5

$\rightarrow$ $5\,z = 9\,x - 2\,y - 13$

(5z = 9x - 2y - 13)-5z

$\rightarrow$ $0 = 9\,x - 2\,y - 5\,z - 13$

Tason normaalimuotoinen yhtälö on 9x-2y-5z-13=0

Tarkistus

A:=oa

$\rightarrow$ $A := \begin{pmatrix} 1 \\ 3 \\ -2 \end{pmatrix}$

B:=oa+s_1

$\rightarrow$ $B := \begin{pmatrix} 3 \\ 2 \\ 2 \end{pmatrix}$

C:=oa+s_2

$\rightarrow$ $C := \begin{pmatrix} 4 \\ 4 \\ 3 \end{pmatrix}$

TA:=Taso[A,B,C]

$\rightarrow$ $TA := -9x + 2y + 5z = -13$

Tason yhtälön olisi saanut suoraan piirtämälläkin.